
 Structured Text Blocks

T428_Control
'------------------------------------- 
' Example 12.3 
' 
' Description: Control of Tank T-428 
' 
' Copyright (c) 2015 Dogwood Valley Press, LLC 
'------------------------------------- 

(* Convert analog input to level *)
INT_TO_REAL( IN:=LT428_MEAS, Q=>TmpReal);
LT428_Val := TmpReal * (15.0-1.0)/32000.0 + 1.0;
(* Level Control: turn on when low, turn off when high. *) 
(* If not enabled, always turn off.                     *)
If T428_Cntrl THEN 
    IF (LT428_Val < T428_Min) THEN
        XV427_OPEN := 1; 
    END_IF; 
    IF (LT428_Val > T428_Max) THEN
        XV427_OPEN := 0; 
    END_IF; 
ELSE
    XV427_OPEN := 0; 
END_IF;

T428_OI_Alarms
'------------------------------------- 
' Example 12.4 
' 
' Description: Operator interface and alarm functions for T-428  
' 
' Copyright (c) 2023 Dogwood Valley Press, LLC 
'------------------------------------- 
(* Make sure T428_Min is at least 2.1 and no more than 13.4 *)
IF (T428_Min < 2.1) THEN T428_Min := 2.1; END_IF; 
IF (T428_Min > 13.4) THEN T428_Min := 13.4; END_IF;

(* Calculate max level *)
T428_Max := T428_Min + 1.5;

(* Low level alarms *)
T428_LOLA := (LT428_Val < 4.0);
T428_Hrn_Act := (LT428_Val < 2.0);

(* Transitions for level and ack button *)
HrnTrig(CLK:=T428_Hrn_Act);

AckTrig(CLK:=Alm_Ack);
(* Trigger horn when level drops below 2.0 or *) 
(*  stays below 2 for 5 minutes after ack'ed. *)

IF (HrnTrig.Q) OR (Ack_Tmr.Q) THEN
    T428_HORN := 1; 
ELSE 
    IF AckTrig.Q THEN
        T428_HORN := 0; 
    END_IF; 
END_IF;

(* Time level staying below 2 after ack *)
IF (AckTrig.Q AND T428_Hrn_Act) THEN
    Ack_Tmr_En := 1; 
END_IF; 
IF Ack_Tmr_En AND (NOT T428_Hrn_Act OR Ack_Tmr.Q ) THEN
    Ack_Tmr_En := 0; 
END_IF;
Ack_Tmr( IN:=Ack_Tmr_En, PT:=300000);

